NAG Fortran Library Routine Document

F07AGF (SGECON/DGECON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07AGF (SGECON/DGECON) estimates the condition number of a real matrix A, where A has been factorized by F07ADF (SGETRF/DGETRF).

2 Specification

```
SUBROUTINE F07AGF(NORM, N, A, LDA, ANORM, RCOND, WORK, IWORK, INFO)
ENTRY      sgecon (NORM, N, A, LDA, ANORM, RCOND, WORK, IWORK, INFO)
INTEGER            N, LDA, IWORK(*), INFO
real               A(LDA,*), ANORM, RCOND, WORK(*)
CHARACTER*1        NORM
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number of a real matrix A, in either the 1-norm or the infinity-norm:

$$\kappa_1(A) = \|A\|_1\|A^{-1}\|_1 \quad \text{or} \quad \kappa_\infty(A) = \|A\|_\infty\|A^{-1}\|_\infty.$$

Note that $\kappa_\infty(A) = \kappa_1(A^T)$.

Because the condition number is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine should be preceded by a call to F06RAF to compute $\|A\|_1$ or $\|A\|_\infty$, and a call to F07ADF (SGETRF/DGETRF) to compute the LU factorization of A. The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\|A^{-1}\|_1$ or $\|A^{-1}\|_\infty$.

4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Parameters

1: NORM – CHARACTER*1 *Input*

On entry: indicates whether $\kappa_1(A)$ or $\kappa_\infty(A)$ is estimated as follows:

if NORM = '1' or 'O', $\kappa_1(A)$ is estimated;

if NORM = 'I', $\kappa_\infty(A)$ is estimated.

Constraint: NORM = '1', 'O' or 'I'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

3: A(LDA,*) – ***real*** array *Input*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the LU factorization of A, as returned by F07ADF (SGETRF/DGETRF).

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07AGF (SGECON/DGECON) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

5: ANORM – ***real*** *Input*

On entry: if NORM = '1' or 'O', the 1-norm of the **original** matrix A; if NORM = 'I', the infinity-norm of the **original** matrix A. ANORM may be computed by calling F06RAF with the same value for the parameter NORM. ANORM must be computed either **before** calling F07ADF (SGETRF/DGETRF) or else from a **copy** of the original matrix A.

Constraint: ANORM ≥ 0.0.

6: RCOND – ***real*** *Output*

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, A is singular to working precision.

7: WORK(*) – ***real*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 4 * \mathrm{N})$.

8: IWORK(*) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, \mathrm{N})$.

9: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form $Ax = b$ or $A^T x = b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $2n^2$ floating-point operations but takes considerably longer than a call to F07AEF (SGETRS/DGETRS) with 1 right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The complex analogue of this routine is F07AUF (CGECON/ZGECON).

9 Example

To estimate the condition number in the 1-norm of the matrix A, where

$$A = \begin{pmatrix} 1.80 & 2.88 & 2.05 & -0.89 \\ 5.25 & -2.95 & -0.95 & -3.80 \\ 1.58 & -2.69 & -2.90 & -1.04 \\ -1.11 & -0.66 & -0.59 & 0.80 \end{pmatrix}.$$

Here A is nonsymmetric and must first be factorized by F07ADF (SGETRF/DGETRF). The true condition number in the 1-norm is 152.16.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07AGF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA
      PARAMETER        (NMAX=8,LDA=NMAX)
      CHARACTER        NORM
      PARAMETER        (NORM='1')
*     .. Local Scalars ..
      real             ANORM, RCOND
      INTEGER          I, INFO, J, N
*     .. Local Arrays ..
      real             A(LDA,NMAX), WORK(4*NMAX)
      INTEGER          IPIV(NMAX), IWORK(NMAX)
*     .. External Functions ..
      real             F06RAF, X02AJF
      EXTERNAL         F06RAF, X02AJF
*     .. External Subroutines ..
      EXTERNAL         sgecon, sgetrf
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07AGF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*        Compute norm of A
*
         ANORM = F06RAF(NORM,N,N,A,LDA,WORK)
*
*        Factorize A
*
         CALL sgetrf(N,N,A,LDA,IPIV,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
*
*           Estimate condition number
*
            CALL sgecon(NORM,N,A,LDA,ANORM,RCOND,WORK,IWORK,INFO)
*
            IF (RCOND.GE.X02AJF()) THEN
               WRITE (NOUT,99999) 'Estimate of condition number =',
     +           1.0e0/RCOND
            ELSE
               WRITE (NOUT,*) 'A is singular to working precision'
```

```
            END IF
          ELSE
            WRITE (NOUT,*) 'The factor U is singular'
          END IF
        END IF
        STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END
```

9.2 Program Data

```
F07AGF Example Program Data
  4                             :Value of N
  1.80   2.88   2.05  -0.89
  5.25  -2.95  -0.95  -3.80
  1.58  -2.69  -2.90  -1.04
 -1.11  -0.66  -0.59   0.80    :End of matrix A
```

9.3 Program Results

```
 F07AGF Example Program Results

 Estimate of condition number =  1.52E+02
```